

REFERENCE

NBS
PUBLICATIONS



NBSIR 85-3144

ASET-B A Room Fire Program for Personal Computers

William D. Walton

U.S. DEPARTMENT OF COMMERCE
National Bureau of Standards
National Engineering Laboratory
Center for Fire Research
Gaithersburg, MD 20899

April 1985

Supported in part by:
National Park Service
Washington, DC

and

Department of Health and Human Services
Washington, DC

QC
100
.U56
85-3144
1985

Ref-NRS

OC100

.U56

85-3144

1985

NBSIR 85-3144

**ASET-B A ROOM FIRE PROGRAM FOR
PERSONAL COMPUTERS**

William D. Walton

U.S. DEPARTMENT OF COMMERCE
National Bureau of Standards
National Engineering Laboratory
Center for Fire Research
Gaithersburg, MD 20899

April 1985

Supported in part by:
National Park Service
Washington, DC

and

Department of Health and Human Services
Washington, DC



U.S. DEPARTMENT OF COMMERCE, Malcolm Baldrige, *Secretary*
NATIONAL BUREAU OF STANDARDS, Ernest Ambler, *Director*

TABLE OF CONTENTS

| | Page |
|---|------|
| LIST OF FIGURES | iv |
| Abstract | 1 |
| 1. INTRODUCTION | 2 |
| 2. DESCRIPTION OF THE MODEL | 4 |
| 3. SOLUTION OF THE EQUATIONS | 7 |
| 4. STRUCTURE OF THE COMPUTER PROGRAM | 10 |
| 4.1 General Structure | 10 |
| 4.2 Program Sections | 11 |
| 4.2.1 Input and Initialization Section | 11 |
| 4.2.2 Equation Solving Section | 13 |
| 4.2.3 Subroutine to Calculate Heat Release Rate | 13 |
| 4.2.4 Output Subroutine | 14 |
| 5. RUNNING THE PROGRAM | 15 |
| 5.1 General Instructions | 15 |
| 5.2 Program Inputs | 16 |
| 5.2.1 Title | 16 |
| 5.2.2 Heat Loss Fraction | 16 |
| 5.2.3 Height if the Base of the Fire | 17 |
| 5.2.4 Room Ceiling Height and Floor Area | 18 |
| 5.2.5 Maximum Time | 18 |
| 5.2.6 Heat Release Rate | 18 |
| 5.3 Program Outputs | 21 |
| 6. USE AND LIMITATIONS OF THE ASET-B RESULTS | 21 |
| 6.1 Evaluating Fire Safety Objectives | 21 |
| 6.2 Using ASET-B for Multiple Rooms | 23 |
| 6.3 Program Limitations | 23 |
| 6.4 Program Verification | 25 |
| 7. TRANSPORTING THE PROGRAM | 25 |
| 8. ACKNOWLEDGEMENTS | 27 |
| 9. REFERENCES | 27 |
| APPENDIX A: ASET-B PROGRAM LISTING | 29 |
| APPENDIX B: VARIABLE NAMES USED IN ASET-B | 31 |
| APPENDIX C: SAMPLE RUN OF ASET-B PROGRAM | 34 |

LIST OF FIGURES

| | Page |
|---|------|
| Figure 1. Schematic of fire phenomena | 28 |

ASET-B A ROOM FIRE PROGRAM FOR PERSONAL COMPUTERS

William D. Walton

ABSTRACT

ASET-B, a personal computer program for predicting the fire environment in a single room, is presented. ASET-B solves the same differential equations as the previously developed computer program, ASET (Available Safe Egress Time), using a simpler numerical technique. ASET-B requires as input the height and area of the room, the elevation of the fire above the floor, a heat loss factor, and a fire specified in terms of heat release rate. The program predicts the thickness and the temperature of the hot smoke layer as a function of time. ASET-B is written in BASIC and is not subject to copyright. This paper describes the program and its use. Included are a listing of the program, program variable name listing and a sample run. A discussion of user modifications also is given.

Key Words: compartment fires; computer programs; evacuation; fire growth; manuals; mathematical models; room fires; smoke.

1. INTRODUCTION

In 1980 L. Y. Cooper of the Center for Fire Research, National Bureau of Standards introduced ASET, a mathematical model for estimating Available Safe Egress Time in fires [1]¹. In 1982, Cooper with D. Stroup published a computer program to perform the calculations in the mathematical model [2], thus, the computer program also became known as ASET. ASET was not specifically written for the personal computer environment because at the time it was being developed, personal computers were just emerging as a tool for use in the engineering office. Since the introduction of ASET, the use of personal computers has become widespread and there has been significant interest in running ASET on personal computers. In response to this interest, this paper introduces ASET-B, a program for personal computers based on the original ASET mathematical model. The B is used to indicate basic, brief, BASIC and Beta.

Most larger personal computers, when properly configured, can run the original ASET program. There are however, several reasons why ASET is not the most appropriate program for the personal computers widely used today. ASET is written in FORTRAN and consists of over 1500 program lines. Although FORTRAN is available on many personal computers, BASIC is by far the most popular programming language. To make ASET-B easy to distribute

¹ Figures in brackets indicate literature references at the end of this report.

and allow the user to make modifications, it has been written in BASIC. A second reason that ASET may not be appropriate for some personal computers, is its size and run time. ASET has a wide variety of features which allow the user to run multiple cases, identify endpoints and calculate species concentrations. These features increase program size, input, output and run time. In developing ASET-B it was decided to make the program as simple and fast as possible, while providing the same essential output information produced by ASET. While ASET-B can be used to address a wide variety of problems, it can also serve as an introduction to the more comprehensive ASET.

ASET-B as a program looks very different from the original ASET. ASET-B consists of approximately 100 program lines in BASIC. Most of the variable names have been changed in order to make the program more compact and easier to follow. The multiple run options have been eliminated as well as the ability to calculate species concentrations and automatically identify endpoints. With these exceptions, the primary equations solved by ASET and ASET-B are identical. The most significant change in ASET-B is the use of a different mathematical procedure to solve the primary equations. ASET-B employs an equation solver that is at least 5 times faster than that used in ASET, while retaining mathematical agreement to within a fraction of a percent. It reduces the time for an average run on a typical 16 bit personal computer from over 5 minutes to under 1 minute.

2. DESCRIPTION OF THE MODEL

The mathematical model which is the basis for ASET and thus ASET-B has been presented in detail by Cooper [1,3], and will be only summarized here. It is based on a single room or enclosure with all doors, windows or vents closed except for a small leak at floor level. This leak prevents the pressure from increasing in the room. A fire starts at some point below the ceiling and releases energy and products of combustion. The rate at which energy and products of combustion are released may change with time. The hot products of combustion form a plume, which due to buoyancy, rises towards the ceiling. As the plume rises, it draws in cool air from the room which decreases the plume's temperature and increases its volume flow rate. When the plume reaches the ceiling it spreads out and forms a hot gas layer, which descends with time as the plume's gases continue to flow into it. There is a relatively sharp interface between the hot upper layer and the air in the lower part of room, which in this model is considered to be at ambient temperature. The only interchange between the air in the lower part of the room and the hot upper layer is through the plume. ASET could therefore be described as a two layer or zone model. The basic fire phenomena are shown schematically in figure 1.

The two unknowns in ASET-B are the height of the hot layer interface above the fire, Z , and the average temperature of the upper layer, P . It should be noted that the notation used here to describe the model is consistent with the variable names used

in the computer program. The unknowns, Z and P , are often referred to as the (dimensionless) height and temperature of the smoke layer since, consistent with the model formulation, smoke can only be found in the plume and the hot upper layer. The known quantities are the area and height of the room, A and H , the height of the base of the fire above the floor, F , and the acceleration due to gravity, G . In addition, the ambient temperature, P_A , density, D_A , and specific heat, C_P , of air must be known. The final known quantities are the rate at which heat is released by the fire as a function of time, Q_T , the fraction of the total heat release which is given off as radiation, L_R , and the fraction of total heat release rate which is lost to the contents and surrounding surfaces of the room, L_C .

The unknown height and temperature are determined by using conservation of mass and energy in conjunction with equations describing the plume. Since the height and temperature of the smoke layer will vary with time, T , their solutions are obtained by solving two differential equations. In developing the original equations for ASET [1,3], two dimensionless groups of problem parameters, C_1 and C_2 , were introduced. Also introduced were dimensionless forms of the variables time, height and temperature of the smoke layer, initial height of the smoke layer, height of the base of the fire and the rate of heat release. These variables are made dimensionless by dividing them by a characteristic quantity with the same dimensions or units. Thus the dimensionless temperature, P , is the actual temperature

of the smoke layer, P_F (converted to $^{\circ}R$), divided by the ambient temperature, P_A ($^{\circ}R$). Similarly, the dimensionless rate of heat release, Q_T , is the actual rate of heat release, Q_A (kW), divided by the initial rate of heat release, Q_0 (kW). Finally, the dimensionless variables, height of the smoke layer, Z , initial height of the smoke layer, Z_0 , and height of the base of the fire, F , are the dimensional values for these variables in feet divided by a characteristic length CL which is also in feet. Here as in the ASET program, CL is simply taken as one foot. Thus the dimensionless lengths Z , Z_0 and F are the same as their physical lengths in feet. The dimensionless time, T , is the actual time divided by a characteristic time, CT , of one second. The dimensionless time, T , is therefore numerically equal to the actual time in seconds. Since engineering units are used in ASET, this convention has been continued here for consistency. Conversion to SI units is provided in the computer program.

The differential equations for the dimensionless height of the layer above the fire, Z , and average temperature of the layer, P , are given below.

$$\frac{dZ}{dT} = \begin{cases} -C_1 \cdot Q_T - C_2 \cdot Q_T^{1/3} Z^{5/3} & 0 < Z \leq Z_0 \\ -C_1 \cdot Q_T & -F < Z \leq 0 \\ 0 & Z = -F \end{cases}$$

$$\frac{dP}{dT} = \begin{cases} P [C_1 \cdot Q_T - (P-1) C_2 \cdot Q_T^{1/3} Z^{5/3}] / (Z - Z_0) & 0 < Z < Z_0 \\ P \cdot C_1 \cdot Q_T / (Z_0 + Z) & -F \leq Z \leq 0 \end{cases}$$

$$C_1 = (1 - LC) \cdot Q_0 \cdot CT / (DA \cdot CP \cdot PA \cdot A \cdot CL)$$

$$C_2 = (0.21 \cdot CT / A) [(1 - LR) \cdot Q_0 \cdot G \cdot CL^2 / (DA \cdot CP \cdot PA)]^{1/3}$$

In order to solve the equations for Z and P the initial conditions must be known. One set of initial conditions which were derived in [1,3], and will be used here, assume that the fire starts with a small heat release rate, Q_0 , at time $T=0$. Under such conditions the initial conditions are.

$$Z=Z_0$$

$$P=1+Z_0^{-5/3} C_1/C_2$$

Although dP/dT is indeterminate in the above equation at $T=0$ its actual value has been found in [1,3] to be.

$$dP/dT=(C_1/C_2)[2 \cdot DQ_0+5(C_1+C_2 \cdot Z_0^{5/3})]/(6 \cdot Z_0^{8/3}),$$

Where $DQ_0=dQ_T/dT$ at time $T=0$.

3. SOLUTION OF THE EQUATIONS

In general, the differential equations for Z and P cannot be solved explicitly, that is, an algebraic expression cannot be written which describes Z and P at any time T. As a result, the equations must be solved numerically. ASET solves the differential equations using a variation of the fourth-order Runge-Kutta method with variable time step. While this method has a high degree of accuracy, it has been determined that the improved Euler's method has sufficient accuracy for this problem. The improved Euler's method is a simple predictor-corrector type and is described in most books on numerical methods [4]. The improved Euler's method used in ASET-B requires substantially fewer calculations than the method used in ASET, resulting in ASET-B running much faster than ASET.

The improved Euler's method as applied in ASET-B is basically a technique for stepping the solution forward in time. Given the values of Z and P at a particular time, T, the method is used to determine the values of Z and P at time, T+DT, where DT is a small time increment. This process is started at time T=0 and continued until Z and P are known at all times of interest. In the case of ASET-B an increment of 1 second has been found to yield results which agree well with ASET for problems of practical interest.

In ASET-B, Z1 and P1, are used to indicate the values of Z and P at time, T. For the first step these are the initial values at time, T=0. Z2 and P2 are used to indicate the values of Z and P to be calculated at time, T+DT. To determine Z2 and P2 it is observed that the differential equations for Z and P represent the time rate of change of these quantities. The time rate of change multiplied by the time step yields the change which occurs over the time step. This would be an exact result if the equations were linear or the time steps were infinitely small. Since the equations are nonlinear, and it is impractical to make the time step infinitely small, an approximation must be used. In the improved Euler's method, Z2 and P2 are first predicted using the derivatives evaluated at time, T. Using Z2 and P2 the derivatives are then evaluated at time, T+DT. Corrected values of Z2 and P2 are then calculated using the average of the derivatives evaluated at times T, and, T+DT. Z2 and P2 are predicted by

$$Z2=Z1+DZ1 \cdot DT$$

$$P2=P1+DP1 \cdot DT$$

where $DZ1=dZ/dT$ and $DP1=dP/dT$ are evaluated using $Z=Z1$ and $P=P1$. The derivatives at time $T+DT$, $DZ2=dZ/dT$ and $DP2=dP/dT$, are then evaluated using $Z=Z2$ and $P=P2$. Corrected values for $Z2$ and $P2$ are calculated using the average derivatives.

$$Z2C=Z1+[(DZ1+DZ2)/2] \cdot DT$$

$$P2C=P1+[(DP1+DP2)/2] \cdot DT$$

The predicted values of Z and P are then compared to the corrected values. In ASET-B if the absolute value of the difference between the predicted and corrected values is less than 0.001, the solution is considered to have converged and the program proceeds to the next time step. If the difference is greater than 0.001 the predicted values become the corrected values and the derivatives at time, $T+DT$, are recalculated. New corrected values are then calculated. In ASET-B this procedure is repeated for a maximum of thirty times. If the differences are still greater than 0.001, a warning is printed, and the program proceeds to the next time step.

The evaluation of the derivatives of Z and P requires the dimensionless heat release rate, QT , be known for all times, T . For heat release rates which are not constant with time, ASET-B requires the heat release be specified for each one second time interval. To simplify this procedure, ASET-B uses point specified heat release rates with linear interpolation. Heat release rates

can be specified at as many as 100 different times. Linear interpolation is then performed to determine the heat release rate at each time step.

4. STRUCTURE OF THE COMPUTER PROGRAM

4.1 General Structure

ASET-B is structured to be easy to understand and modify, while being adaptable to a variety of computers. In some cases these objectives conflict since BASIC is not a structured programming language and the implementation of BASIC varies among computers. As a result, some of the so called rules of structured programming have not been followed in favor of a program that can be adapted to most forms of the BASIC programming language. A listing of the program is given in appendix A and an index of variable names in appendix B. In order to make the listing compact, all documentation is provided in the text of this paper.

ASET-B is designed to be easy to modify as well as easy to use. It is intended to serve as a base program which the user can modify to meet a particular set of needs. Although the number of possible user modifications is almost unlimited, comments on several typical modifications have been provided with the program description. If user modifications are made to the program, a comment should be added to the header printed in lines 10 and 440. This will indicate that the program has user modifications and is not the standard program.

4.2 Program Sections

The program consist of four major sections. The first, lines 10-530, reads and writes the required input data, sets initial parameters and performs initial calculations. The second, lines 540-870, consist of the calculations required to step the solution through time. Finally, lines 880 to 960 are a subroutine which calculates the rate of heat release of the fire, and lines 970 to 1080 a subroutine which writes results. A brief description of each of the sections is provided below.

4.2.1 Input and Initialization Section

The first section of the program can be subdivided by function. The input data is read, and unit conversions performed, in lines 10-250. The units of input lengths and area are feet and square feet respectively with conversion to meters and square meters in lines 110-130. The units of these inputs may be changed to meters and square meters by reversing the input variable names and conversion factors in lines 80-130. In line 60, the fraction of the heat release rate of the fire which is instantaneously lost by radiation for the combustion zone, LR, is set equal to 0.35. A discussion of this parameter which is called ALAMR in ASET can be found in [1].

The input heat release rates are read and stored in lines 150-250. The rate of heat release at time zero, Q0, has been set

at 0.1 kilowatts in line 150, corresponding to an initial fire size which approaches zero. This value can be increased by the user if needed. ASET-B can accommodate 100 pairs of input times and heat release rates. While this number should be adequate for most cases it can be changed by substituting the new number for 100 in lines 40, 170 and 230. The units for the input heat release rates are kilowatts and may be changed to other units by placing conversion factors in lines 200 and 210.

Lines 260-420 initialize variable values and perform initial calculations. Several of the variables defined in these lines may be modified to suit the user. This may be accomplished by either changing the values within the program or by creating new prompt questions and assigning the variables with input values. In lines 260 and 270 the ambient temperature, PA, and density, DA, of air may be changed to reflect an initial temperature other than 70 °F. In line 410 the time interval for printing, OI, may be changed from its default value of 5 seconds. The results are calculated for every second but are printed only at the time interval specified by OI.

Lines 430-530 write out a summary of the input data, the column headings for the subsequent output and the initial output. Some users may desire this output to be sent to a printer or disk file in addition to, or in place of, the output to the screen. The techniques for accomplishing this vary with computer but the most common are the use of LPRINT and PRINT#. Similar techniques may be used to read the input data from a disk file.

4.2.2 Equation Solving Section

The second section of the program, lines 540-870, solves the differential equations following the methodology outlined earlier in this report. This section consists of two primary loops. The outer loop beginning at line 540 and ending at line 860 corresponds to a single time step. The inner loop beginning at line 620 and ending at line 800 is repeated only if the absolute value of the difference between the predicted and corrected values of either Z or P is greater than 0.001. This comparison is made in line 750.

At the beginning of the outer loop are two checks to determine if the solution process should be terminated. In line 560 the current time, T2, is compared to the maximum input time, TL, and in line 570 to the last time for which a heat release rate is specified, TQ(NQ). If the current time is greater than either of these two time limits the run is terminated with transfer to line 870.

4.2.3 Subroutine to Calculate Heat Release Rate

The third section of the program consist of a subroutine , lines 880-960, which calculates the rate of heat release of the fire and is called from the outer calculation loop at line 580. This subroutine utilizes the input rates of heat release at

specified times and performs linear interpolation to determine the rate of heat release at each time step. Although this method of specifying the heat release rate is the most general, the user may wish to modify the program to accept other models of fire heat release. The ASET program for example can accommodate multi-point exponential fire growth input. Other fire growth models can be inserted in place of, or in addition to, the existing point specified model. This is accomplished by modifying the input section in lines 160-250 and subroutine in lines 880-960.

4.2.4 Output Subroutine

The second subroutine, lines 970-1080, is called from the outer calculation loop at line 850. This subroutine converts the dimensionless variables to ones with units and writes the results. The particular variables printed in this subroutine have been selected to satisfy the needs of most users. The speed of program execution may be affected by the quantity of the output, particularly if a printer is used. As a result, the user may want to reduce the quantity of output to reduce the run time. This is accomplished by eliminating the variables to be printed in line 1070 and the corresponding column headings in lines 510 and 520. There are two values calculated in this subroutine which have not been printed. These are the rate of rise of temperature in the upper layer in degrees Fahrenheit, RRF, and Celsius, RRC, per minute. These values may be printed by adding their variable names to the output list and including the appropriate column

headings. It should be noted that rate of rise of the upper layer is not the same as the rate of rise in a heat detector. The temperature of the heat sensing element is a function of the heat transfer between the hot gas and the element.

ASET-B produces only tabular output and has no specific provisions for graphics. The reason is that graphics commands are generally dependent on the device which produces the graphics. There are however several methods by which graphics can be added to ASET-B. The user may write the tabular results to a file which can be read as input by a commercial graphics routine. The results can then be presented using the graphics commands in the commercial routine. Alternatively, the user may write a graphics routine into the ASET-B program.

5. RUNNING THE PROGRAM

5.1 General Instructions

ASET-B is written as an interactive program, that is, the program prompts the user with questions. The mechanics of answering the questions is very straightforward, however the selection of the input data requires further explanation. The procedure for loading and running the program will not be covered in this paper, but can be found in the user's guide included with the computer. When the program is started it will initially print the program name, a version number and a copyright notice. The program will then print the first of seven questions and wait for

a response to that question. After the user has answered a question, a return or enter key is typed to register the answer and proceed to the next question. No special error checking is provided in the program so the user should carefully verify all entries. If an error is detected after the data has been entered, the program must be terminated and restarted.

5.2 Program Inputs

5.2.1 Title

The first prompt question or input is a run title. This title is used to identify the run and may consist of characters and/or numbers. The title will be printed at the beginning of the output.

5.2.2 Heat Loss Fraction

The second input is the heat loss fraction. This quantity is the instantaneous fraction of the heat release rate of the fire which is lost to the bounding surfaces of the room and its contents. Cooper has provided guidelines for selecting this parameter which is called Lamda C or ALMAC in ASET [1,3]. He has determined that the approximate range is 0.6 - 0.9. The lower value corresponds to high aspect ratio spaces (ratio of ceiling span to room height) with smooth ceilings and fires positioned far away from the walls. The intermediate to high values

correspond to low aspect ratio spaces, rooms with irregular surfaces or rooms in which the fire is within one ceiling height of the wall. The temperature of the upper layer is a function of the heat loss fraction and the heat release rate of the fire. The greater the heat loss fraction the lower the temperature in the upper layer. The heat loss fraction for a room with insulated walls will be lower than the fraction for the same room with uninsulated walls.

Both ASET and ASET-B treat the heat loss parameter as a constant. That is, the heat lost from the room is a constant fraction of the heat release rate of the fire. As the heat release rate of the fire changes the quantity of heat lost will also change but in direct proportion to the fire. Therefore the room will not cool down even though the heat release rate of the fire goes to zero.

5.2.3 Height of the Base of the Fire

The third input is the height of the base of the fire above the floor in feet. For fuel items of relatively uniform surface height, such as beds, this is simply the height of the surface. For three dimensional fuel items, such as sofas, an average height weighted to reflect the distribution of surfaces should be used. The rate of growth of the upper layer is strongly dependent on the difference between the height of the base of the fire and the height of the smoke layer interface.

5.2.4 Room Ceiling Height and Floor Area

The fourth and fifth inputs are the room ceiling height in feet and the floor area in square feet. According to references [1,3] the calculations may not be valid when applied to room length-to-width aspect ratios greater than 10:1, or with a ratio of height to minimum horizontal dimension exceeding one. The equations are based on the assumption that the upper layer is well mixed and at a uniform temperature. Therefore the results for a square room and a rectangular room of equal height and area will be the same.

5.2.5 Maximum Time

The sixth input is the maximum time for the simulation in seconds. The results of the calculations will be printed at five second intervals until the maximum time or until the end of the heat release data.

5.2.6 Heat Release Rate

The final input is the description of heat release rate of the fire. As described earlier the program can accommodate up to 100 pairs of times and corresponding heat release rates. The program performs a linear interpolation between the specified points to determine the heat release rates at the required times

during the calculations. The data is entered by typing the time in seconds, followed by a comma, followed by the heat release rate in kilowatts. A return or enter is then typed to proceed to the next line. Heat release rates entered as less than 0.1 kilowatts will be converted to that value. The program will automatically assume a starting value 0.1 kilowatts at time zero. A heat release rate at time zero does not have to be entered unless a greater initial heat release rate is required. When all of the desired times and heat release rates have been entered a -9,-9 followed by a return is entered to terminate the data entry and begin the calculations. Actually any negative time followed by a heat release rate will result in the same action.

Heat release data for some furniture items can be found in [5]. These data can be entered directly by selecting a number of the points which represent the general shape of the curve. Equations for the growth rate of a number of industrial commodities can be found in NFPA 72E and 204M [6,7]. These equations can be solved to yield the heat release rate at selected times, which can then be entered into the program. In addition, correlations of pool and crib fires have been summarized in appendix B of reference [8].

In many cases there is insufficient knowledge of the fuel or no data on which to determine a heat release rate for the fuel. One possibility in this case is to use the so called semi-universal fire developed by Cooper [1,3]. Most fires, other than

rapid growth fires such as flammable liquids, have heat release rates less than the semi-universal fire. The semi-universal fire is based on the same Factory Mutual data used in NFPA 204M and 72E . Initially it simulates a fire on a polyurethane mattress, at 148 seconds it begins to simulate burning wood and polyurethane pallets, and at 349 seconds a fire spreading through commodities in paper cartons and stacked on pallets. The mathematical description of the semi-universal fire is provided in appendix C. The semi-universal fire will generally provide conservative results when used in ASET-B to predict the height and temperature of the upper layer. If the semi-universal fire is used to predict time to detection, it may predict detection too quickly.

ASET-B cannot change the burning rate of the fire as a function of the conditions in the room. ASET-B cannot determine if there is sufficient oxygen present for the fire to burn as prescribed and it cannot increase the burning rate as a result of thermal radiation to the fuel. The burning rate data referenced above is based on free burn conditions, that is with a free supply of ambient air and no thermal enhancement from the surroundings. The use of free burn data should prove satisfactory at least up to the time when ASET-B predicts upper layer temperatures of approximately 350-450°C. Beyond that time ASET-B will continue to calculate layer heights and temperatures based on the input fire even though the results may be physically impossible.

5.3 Program Outputs

The output of the ASET-B program is a summary of the input data and a table of the conditions in the room as a function of time. The first column in the table is the simulation time in seconds. The second and third columns are the temperature in the upper layer in degrees Fahrenheit and Celsius. The fourth and fifth columns are the height above the floor of the interface between the upper and lower layers. The sixth and seventh columns are the heat release rate of the fire in kilowatts and BTU per second. Because of the formatting limitations in the BASIC language, some columns may show more significant digits than are meaningful.

6. USE AND LIMITATIONS OF THE ASET-B RESULTS

6.1 Evaluating Fire Safety Objectives

Fire simulation computer programs such as ASET-B generally provide a description of the environment within an enclosure and not a direct evaluation of the fire safety. While a complete discussion of the use of these programs in evaluating fire safety is beyond the scope of this paper a few examples will be given.

A common objective in fire safety design is to determine the time available for safe egress. In fact, ASET was originally written with this objective in mind [9]. Although the design objective may be stated in different ways, one such statement is

as follows. Given a particular room and fire, how long does it take to reach untenable conditions within the room? To evaluate this objective, the limits of tenability must be established in terms of the program results. For example, the room will become untenable when the smoke layer reaches head height, or when the temperature in the layer reaches a particular value. Using these tenability limits and the program results, the time to untenable conditions can be evaluated.

A second common fire safety design objective is to determine if equipment or products will sustain damage in the fire environment. Equipment damage thresholds can be established in the same manner as human tenability limits. The damage thresholds can be compared to the predictions to determine if equipment will be damaged and the time to the onset of damage. The results can then be evaluated based on estimates of detection time, suppression effectiveness and salvage capability.

In addition to determining the time to untenable conditions or damage, ASET-B can be used to evaluate the impact of changing one or more of the input variables. For example, the sensitivity of the resultant fire environment to room size can be examined by making multiple runs with different room sizes while holding the other input values constant.

6.2 Using ASET-B For Multiple Rooms

Although the ASET-B program is based on a single room with closed doors and windows, it is possible to use the program to make approximations for rooms with open doors or windows or even multiple rooms. The results of the closed room ASET- program are generally valid for rooms with open doors or windows until the interface drops to the top of the door or window. For the case of a door opening into another closed room, a second run can be made using the area of the two rooms combined. The first run (with floor area of single room) is used to predict conditions until the interface drops to the top of the door. The second run (with floor area of combined rooms) is used from the time the newly computed interface drops to the top of the door until the end of the simulation. Although there will be a time gap between the two runs, this technique will provide satisfactory results in many cases, especially where door openings are relatively large and where fire heat release rates are relatively small.

6.3 Program Limitations

The use of ASET-B or any design aid requires the design engineer to make the final evaluation as to the appropriateness of the design. ASET-B is based on certain engineering approximations of the fire environment and should be used to supplement rather than replace sound engineering judgement. The program results should be treated as approximate and the user is encouraged to become familiar with how changes in the input

variables effect the program results. The temperature of the upper layer and the height of the interface respond differently to changes in the input data. Appropriate factors of safety should be applied to either the input data or the program results.

Some of the limitations of the program have been presented in conjunction with the input data requirements. There are however, some additional limitations. The mathematical procedure used in ASET-B is very hardy, that is, the procedure will normally converge and produce results. There are combinations of input data for which the program will either fail to converge or halt due to an illegal mathematical operation. If the procedure for solving the equations fails to converge, a warning will be printed and the solution will continue. The results following this message may be in error and should be treated as such. The failure to converge is usually a result of a heat release value which changes too rapidly. In most cases this problem can be corrected by minor smoothing of the input heat release curve.

There are combinations of input data which will result in the program attempting to execute mathematical operations which are invalid in the BASIC language. These operations include dividing by zero and raising numbers to very small or large powers. Operations of this type will be flagged by the computer and subsequent results should be considered in error. Invalid operations are usually caused by errors in input data. ASET-B

does not check the input data for possible errors which can cause illegal machine operations or produce unrealistic results such as negative temperatures or interface heights above the ceiling. As indicated previously ASET-B may predict physically impossible temperatures if very large fires are placed in small rooms.

6.4 Program Verification

Results of the ASET program have been compared to data from a limited number of actual fire experiments [1,3]. These comparisons can be extended to the ASET-B program since it produces results which are within a few percent of those produced by ASET. The fire experiments considered a mockup of a hospital room-corridor building space. Comparisons were found to be generally favorable. This does not necessarily mean that the comparison will be favorable in all cases. Clearly additional studies are required in this area and that work is ongoing.

7. TRANSPORTING THE PROGRAM

An important aspect of using any computer program is obtaining a copy of the program for the computer where it is to be used. The program must not only run, but it must produce the correct results. Although a listing of ASET-B is given in appendix A, it is not recommended the user attempt to type in the program. A typographical error is generally very difficult to track down and may result in a program error that occurs only with certain combinations of input data. Instead the user should

obtain the program in machine readable form. Since ASET-B is in the public domain it may be freely copied.

A sample run of the ASET-B program is shown in appendix C. Although the program cannot be thoroughly tested by comparing the output of a single run, this example has been selected to exercise most of the computer program. Results from a copy of the program which do not compare exactly with this sample output probably indicate the program is in error.

The output of the ASET-B program will not agree exactly with the output of the ASET program. This results from the difference in the mathematical procedures used. Also, the ASET program listed in [2] contains an error which causes the calculated temperatures to be approximately one percent too high.

ASET-B was written for Microsoft² interpreted and compiled BASIC and has run successfully on IBM³ and compatible personal computers operating under MS-DOS². ASET-B has not yet been tested on other personal computers, but it should run with no difficulties.

² Microsoft and MS are trademarks of Microsoft Corporation and do not represent an endorsement by the National Bureau of Standards

³ IBM is a trademark of International Business Machines Corporation and does not represent an endorsement by the National Bureau of Standards

8. ACKNOWLEDGEMENTS

Appreciation is extended to Dr. Leonard Cooper and Mr. David Stroup for their assistance in preparing this report.

9. REFERENCES

- [1] Cooper, L.Y., Estimating Safe Available Egress Time from Fires. Nat. Bur. Stand. (U.S.) NBSIR 80-2172; 1981 February. 72 p.
- [2] Cooper, L.Y.; Stroup, D.W. Calculating Safe Egress Time (ASET) - A Computer Program and User's Guide. Nat. Bur. Stand. (U.S.) NBSIR 82-2578; 1982 September. 137 p. also to appear in Fire Safety Jour.
- [3] Cooper, L.Y. A Mathematical Model for Estimating Available Safe Egress Time in Fires. Fire Mater. 6(3/4): 135-144; 1982 September/December.
- [4] Carhanan, B.; Luther, H.A.; Wilkes, J.O. Applied Numerical Methods. New York: John Wiley & Sons; 1969. 604 p.
- [5] Lawson, J.R.; Walton, W.D.; and Twilley, W.H. Fire Performance of Furnishings As Measured in the NBS Furniture Calorimeter. Part 1. Nat. Bur. Stand. (U.S.) NBSIR 83-2787; 1983 August. 133 p.
- [6] Standard on Automatic Fire Detectors 72E. Boston: Nat. Fire Prot. Assn.; 1984.
- [7] Guide for Smoke and Heat Venting 204M. Boston: Nat. Fire Prot. Assn.; 1982.
- [8] Quintiere J.G.; Baum H.R.; Lawson J.R. Fire Growth In Combat Ships. Nat. Bur. Stand. (U.S.) NBSIR; To be published.
- [9] Cooper L.Y. A Concept for Estimating Available Safe Egress Time in Fires. Fire Safety Jour. 5(2): 135-144; 1983.




Figure 1. Schematic of fire phenomena

APF A: ASET-B PROGRAM LISTING

```

10 PRINT"ASET-B VERSIO WRITTEN BY W. D. WALTON 1985"
20 PRINT"CONTRIBUTION NATIONAL BUREAU OF STANDARDS (U. S.)."
30 PRINT"NOT SUBJECT TOIGHT."
40 DIM TQ(100),Q(100)
50 INPUT"ENTER RUN TITLES$
60 LR=.35
70 INPUT"HEAT LOSS FRA(LC
80 INPUT"FIRE HEIGHT (1
90 INPUT"ROOM CEILING H(ft)"; H
100 INPUT"ROOM FLOOR A(ft)"; A
110 FM=F*.3048
120 HM=H*.3048
130 AM=A*.0929
140 INPUT"MAXIMUM TIME"; TL
150 Q0=.1
160 PRINT"INPUT FIRE TND HEAT RELEASE RATES (use -9,-9 to end input)"
170 FOR N=1 TO 100
180 INPUT"TIME (sec), ELEASE RATE (kW)"; TQ(N), QI
190 IF TQ(N) < 0 GOTO
200 IF QI < Q0 THEN Q(
210 Q(N)=QI/Q0
220 NEXT N
230 PRINT"MAXIMUM OF 10T VALUES REACHED"
240 NQ=N-1
250 IF TQ(1)<.1 THEN T1
260 PA=530!
270 DA=.0735
280 CP=.24
290 C1=(1! -LC)*Q0/(DA*A*1.054)
300 C2=.21/A*((1! -LR)^2/(DA*CP*PA*1.054))^(1!/3!)
310 DQ0=(Q(1)-1!)/TQ(1
320 Z0=H-F
330 Z1=Z0
340 DZ1=-C1-C2*Z0^(5!
350 P1=1! +C1/C2*Z0^(-1
360 DP1=C1/C2*(2! ^DQ0+C2*Z0^(5!/3!)))/(6! ^Z0^(8!/3!))
370 T1=0!
380 T2=0!
390 DT=1!
400 QT=1!
410 OI=5!
420 OT=0!
430 PRINT" "
440 PRINT"ASET-B VER O"
450 PRINT TITLES$
460 PRINT"HEAT LOSS FN="; LC
470 PRINT"FIRE HEIGHTt "; FM; "m"
480 PRINT"ROOM HEIGHTt "; HM; "m"
490 PRINT"ROOM AREA=" ft "; AM; "sq m"
500 PRINT" "
510 PRINT" TIME P TEMP LAYER LAYER FIRE FIRE"
520 PRINT" sec F C ft m kW BTU/s"
530 GOSUB 970

```

```

540 T1=T2
550 T2=T1+DT
560 IF T2>TL THEN PRINT"RUN COMPLETE": GOTO 870
570 IF T2>TQ(NQ) THEN PRINT"END OF INPUT FIRE": GOTO 870
580 GOSUB 880
590 Z2=Z1+DZ1*DT
600 P2=P1+DP1*DT
610 IER=0
620 IF Z2<=0! GOTO 670
630 K=C2*QT^(1/3!)*Z2^(5/3!)
640 DZ2=-C1*QT-K
650 DP2=P2*(C1*QT-(P2-1!)*K)/(Z0-Z2)
660 GOTO 720
670 DP2=(P2*C1*QT)/(Z0-Z2)
680 IF Z2<=-F GOTO 710
690 DZ2=-C1^QT
700 GOTO 720
710 DZ2=0!
720 Z2C=Z1+(DZ1+DZ2)/2!*DT
730 P2C=P1+(DP1+DP2)/2!*DT
740 IF Z2C<-F THEN Z2=-F: Z2C=-F
750 IF ABS(Z2C-Z2)<.001 AND ABS(P2C-P2)<.001 GOTO 810
760 IF IER>30 THEN PRINT"WARNING! SOLUTION DID NOT CONVERGE": GOTO 810
770 Z2=Z2C
780 P2=P2C
790 IER=IER+1
800 GOTO 620
810 Z1=Z2C
820 P1=P2C
830 DZ1=DZ2
840 DP1=DP2
850 GOSUB 970
860 GOTO 540
870 END
880 IF T2>TQ(1) GOTO 910
890 QT=Q(1)-((Q(1)-1!)/TQ(1))*(TQ(1)-T2)
900 RETURN
910 FOR N=1 TO NQ
920 IF T2>TQ(N) GOTO 950
930 QT=Q(N)-((Q(N)-Q(N-1))/(TQ(N)-TQ(N-1)))*(TQ(N)-T2)
940 RETURN
950 NEXT N
960 RETURN
970 IF T2<OT THEN RETURN
980 OT=OT+OI
990 QK=QT*QO
1000 QB=QK*.9485
1010 PF=(P1*PA)-460!
1020 PC=(PF-32!)/1.8
1030 RRF=DP1*31800!
1040 RRC=RRF/1.8
1050 X=Z1+F
1060 XM=X*.3048
1070 PRINT USING"#####. # "; T2; PF; PC; X; XM; QK; QB
1080 RETURN

```

APPENDIX B: VARIABLE NAMES USED IN ASET-B

Quantities without units are dimensionless

| | |
|-----|--|
| A | area of room (ft ²) |
| AM | area of room (ft ²) |
| C1 | dimensionless group 1 |
| C2 | dimensionless group 2 |
| CP | specific heat of air (BTU/lb·°F) |
| DA | density of air (lb/ft ³) |
| DP1 | derivative of upper layer temperature with respect to time at the present time step |
| DP2 | derivative of upper layer temperature with respect to time at the next time step |
| DQ0 | derivative of the rate of heat release with respect to time at the initial time |
| DT | time step |
| DZ1 | derivative of height of upper layer above fire with respect to time at present time step |
| DZ2 | derivative of height of upper layer above fire with respect to time at next time step |
| F | height of base of fire (ft) |
| FM | height of base of fire (m) |
| H | height of room (ft) |
| HM | height of room (m) |
| IER | counts number of predictions-corrections |
| K | intermediate value used in evaluating derivatives |
| LC | fraction of heat release lost to contents and surroundings |
| LR | fraction of heat release radiated from fire |
| N | counts number of input rates of heat release |
| NQ | number of input rates of heat release |
| OI | output time interval (s) |

OT time for next output (s)
 P1 temperature of upper layer at present time step
 P2 predicted temperature of upper layer at next time step
 P2C corrected temperature of upper layer at next time step
 PA ambient temperature ($^{\circ}$ R)
 PC temperature of upper layer ($^{\circ}$ C)
 PF temperature of upper layer ($^{\circ}$ F)
 Q rate of heat release specified at time TQ
 QO rate of heat release at initial time step (kW)
 QB rate of heat release at next time step (BTU/s)
 QI rate of heat release specified at time TQ (kW)
 QK rate of heat release at next time step (kW)
 QT rate of heat release at next time step
 RRC rate of rise of temperature in upper layer ($^{\circ}$ C/min)
 RRF rate of rise of temperature in upper layer ($^{\circ}$ F/min)
 T1 present time step
 T2 next time step
 TITLES run title
 TL limit of simulation time (s)
 TQ time for specified rate of heat release Q (s)
 X height of interface above floor (ft)
 XM height of interface above floor (m)
 Z0 height of interface above fire at initial time step
 Z1 height of interface above fire at present time step
 Z2 predicted height of interface above fire at next time step
 Z2C corrected height of interface above fire at next time step

Variable names used in report but not in program

CL characteristic length (1 ft)
CT characteristic time (1 s)
P temperature of upper layer
Z height of interface above the fire

Note: The symbol 0 is the number zero and the symbol O is the
letter O

APPENDIX C: SAMPLE RUN OF ASET-B PROGRAM

ASET-B VERSION 1.0 WRITTEN BY W.D. WALTON 1985
 CONTRIBUTION OF THE NATIONAL BUREAU OF STANDARDS (U.S.).
 NOT SUBJECT TO COPYRIGHT.
 ENTER RUN TITLE? SAMPLE RUN OF ASET-B
 HEAT LOSS FRACTION? .8
 FIRE HEIGHT (ft)? 1
 ROOM CEILING HEIGHT (ft)? 9
 ROOM FLOOR AREA (sq ft)? 225
 MAXIMUM TIME (sec)? 180
 INPUT FIRE TIMES AND HEAT RELEASE RATES (use -9,-9 to end input)
 TIME (sec), HEAT RELEASE RATE (kW)? 20,40
 TIME (sec), HEAT RELEASE RATE (kW)? 100,200
 TIME (sec), HEAT RELEASE RATE (kW)? 180,500
 TIME (sec), HEAT RELEASE RATE (kW)? -9,-9

ASET-B VERSION 1.0
 SAMPLE RUN OF ASET-B
 HEAT LOSS FRACTION= .8
 FIRE HEIGHT= 1 ft .3048 m
 ROOM HEIGHT= 9 ft 2.7432 m
 ROOM AREA= 225 sq ft 20.9025 sq m

| TIME | TEMP | TEMP | LAYER | LAYER | FIRE | FIRE |
|-------|-------|-------|-------|-------|-------|-------|
| sec | F | C | ft | m | kW | BTU/s |
| 0.0 | 70.3 | 21.3 | 9.0 | 2.7 | 0.1 | 0.1 |
| 5.0 | 74.2 | 23.4 | 8.7 | 2.7 | 10.1 | 9.6 |
| 10.0 | 76.7 | 24.9 | 8.3 | 2.5 | 20.0 | 19.0 |
| 15.0 | 79.3 | 26.3 | 7.8 | 2.4 | 30.0 | 28.5 |
| 20.0 | 81.8 | 27.7 | 7.3 | 2.2 | 40.0 | 37.9 |
| 25.0 | 84.6 | 29.2 | 6.9 | 2.1 | 50.0 | 47.4 |
| 30.0 | 87.5 | 30.8 | 6.5 | 2.0 | 60.0 | 56.9 |
| 35.0 | 90.6 | 32.6 | 6.0 | 1.8 | 70.0 | 66.4 |
| 40.0 | 93.9 | 34.4 | 5.7 | 1.7 | 80.0 | 75.9 |
| 45.0 | 97.5 | 36.4 | 5.3 | 1.6 | 90.0 | 85.4 |
| 50.0 | 101.4 | 38.6 | 5.0 | 1.5 | 100.0 | 94.9 |
| 55.0 | 105.6 | 40.9 | 4.7 | 1.4 | 110.0 | 104.3 |
| 60.0 | 110.0 | 43.3 | 4.4 | 1.3 | 120.0 | 113.8 |
| 65.0 | 114.7 | 46.0 | 4.1 | 1.2 | 130.0 | 123.3 |
| 70.0 | 119.8 | 48.8 | 3.9 | 1.2 | 140.0 | 132.8 |
| 75.0 | 125.2 | 51.8 | 3.6 | 1.1 | 150.0 | 142.3 |
| 80.0 | 130.9 | 55.0 | 3.4 | 1.0 | 160.0 | 151.8 |
| 85.0 | 137.0 | 58.3 | 3.2 | 1.0 | 170.0 | 161.2 |
| 90.0 | 143.5 | 61.9 | 3.0 | 0.9 | 180.0 | 170.7 |
| 95.0 | 150.4 | 65.8 | 2.8 | 0.9 | 190.0 | 180.2 |
| 100.0 | 157.6 | 69.8 | 2.6 | 0.8 | 200.0 | 189.7 |
| 105.0 | 165.5 | 74.2 | 2.5 | 0.8 | 218.8 | 207.5 |
| 110.0 | 174.2 | 79.0 | 2.3 | 0.7 | 237.5 | 225.3 |
| 115.0 | 183.7 | 84.3 | 2.1 | 0.7 | 256.3 | 243.1 |
| 120.0 | 194.0 | 90.0 | 2.0 | 0.6 | 275.0 | 260.8 |
| 125.0 | 205.1 | 96.2 | 1.8 | 0.6 | 293.8 | 278.6 |
| 130.0 | 217.2 | 102.9 | 1.7 | 0.5 | 312.5 | 296.4 |
| 135.0 | 230.1 | 110.1 | 1.5 | 0.5 | 331.3 | 314.2 |

| | | | | | | |
|-------|-------|-------|-----|-----|-------|-------|
| 140.0 | 243.9 | 117.7 | 1.3 | 0.4 | 350.0 | 332.0 |
| 145.0 | 258.7 | 125.9 | 1.2 | 0.4 | 368.8 | 349.8 |
| 150.0 | 274.3 | 134.6 | 1.0 | 0.3 | 387.5 | 367.5 |
| 155.0 | 290.7 | 143.7 | 0.8 | 0.2 | 406.3 | 385.3 |
| 160.0 | 307.9 | 153.3 | 0.6 | 0.2 | 425.0 | 403.1 |
| 165.0 | 325.9 | 163.3 | 0.4 | 0.1 | 443.8 | 420.9 |
| 170.0 | 344.6 | 173.7 | 0.2 | 0.1 | 462.5 | 438.7 |
| 175.0 | 364.2 | 184.5 | 0.0 | 0.0 | 481.3 | 456.5 |
| 180.0 | 384.7 | 195.9 | 0.0 | 0.0 | 500.0 | 474.3 |

RUN COMPLETE

Note: The input heat release rate was selected to exercise various sections of the program and does not necessarily represent an actual fire.

